

# Roots of Two Nonlinear Equations for the HP-67

---

To find the roots for the following two simultaneous nonlinear equations:

$$f(x,y) = 0$$

$$g(x,y) = 0$$

$$\text{let } F_x(x,y) = df(x,y)/dx = (f(x+h_x,y) - f(x,y)) / h_x$$

$$\text{let } F_y(x,y) = df(x,y)/dy = (f(x,y+h_y) - f(x,y)) / h_y$$

$$\text{let } G_x(x,y) = dg(x,y)/dx = (g(x+h_x,y) - g(x,y)) / h_x$$

$$\text{let } G_y(x,y) = dg(x,y)/dy = (g(x,y+h_y) - g(x,y)) / h_y$$

Where,

$$h_x = 0.001 * (1 + |x|)$$

$$h_y = 0.001 * (1 + |y|)$$

To refine the guesses for x and y use the following equations:

$$J = F_y(x,y) * G_x(x,y) - F_x(x,y) * G_y(x,y)$$

$$x = x - (F_y(x,y) * g(x,y) - f(x,y) * G_y(x,y)) / J$$

$$y = y - (G_x(x,y) * f(x,y) - g(x,y) * F_x(x,y)) / J$$

## Algorithm

Input: x, y, TolerX, TolerY, and MaxIter

**Iter = 0**

**Do**

**Iter = Iter + 1**

**f = f(x,y)**

**g = g(x,y)**

**h = 0.001 \* (1+|y|)**

**fx = (f(x+h,y) - f) / h**

```

gx = (g(x+h,y) - g)/h
h = 0.001 * (1+|y|)
fy = (f(x,y+h) - f)/h
gy = (g(x,y+h) - g)/h
J = fy * gx - fx * gy
DiffX = (fy * g - f * gy) / J
DiffY = (gx * f - g * fx) / J
x = x - DiffX
y = y - DiffY
Until (|DiffX| < TolerX and |DiffY| < TolerY) or
      (Iter > MaxIter)
Return Iter, x, y

```

## HP-67 Implementation

### Instructions

Note: Numeric output is designated by red italic text.

<i>Step</i>	<i>Input</i>	<i>Command</i>	<i>Output</i>
1	Move program pointer to label 9	[GTO][9]	
2	Switch to program mode.	[PRGM]	
3	Edit the commands between LBL 9 and RTN to implement your f(x,y) function.		
4	When done switch out of program mode.	[RUN]	
5	Move program pointer to label 8	[GTO][8]	
6	Switch to program mode.	[PRGM]	
7	Edit the commands between LBL 8 and RTN to implement your g(x,y) function.		
8	When done switch out of program mode.	[RUN]	
9	Enter the initial guesses for variables x and y. First enter the guess for variable x.	<i>x</i> [ENTER]	
10	Enter initial guess for y.	<i>y</i> [A]	
11	Enter tolerance limit for x.	<i>TolerX</i> [ENTER]	

<i>Step</i>	<i>Input</i>	<i>Command</i>	<i>Output</i>
12	Enter tolerance limit for y.	TolerY[f][A]	
13	Enter maximum number of iterations and start the iterations.	MaxIter[B]	
14	To view the number of iterations.		iterations
15	To view the root for variable x.	[R/S]	x
16	To view the root for variable y.	[R/S]	y
17	To investigate other possible roots, go to step 9.		
18	To solve for a different functions f(x,y) and/or g(x,y) go to step 1.		

### Example

Note: Since the example uses the existing code for functions f(x,y) and g(x,y) found I labels 9 and 8, we will start with the step that simply runs the programs. The code calculates the values for the following functions:

$$f(x,y) = x^2 + y^3 - 31 = 0$$

$$g(x,y) = x * y - 6 = 0$$

The example uses the following input:

- Initial guess for x is 5.
- Initial guess for y is 5.
- Tolerance for the root of variable x is 1E-8.
- Tolerance for the root of variable y is 1E-8.
- The maximum number of iterations is 55.

<i>Step</i>	<i>Comment</i>	<i>Command</i>	<i>Output</i>
1	Enter 5 for the initial guess for x.	5 [ENTER]	
2	Enter 5 for the initial guess for y.	5 [A]	
3	Enter 1E-8 for the tolerance limit for x.	1[EEX]8[CHS][ENTER]	
4	Enter 1E-8 for the tolerance limit for y.	1[EEX]8[CHS][f][A]	
5	Enter 55 for the maximum number of iterations and start the iterations.	55[B]	
6	The program displays the		7.00000

<i>Step</i>	<i>Comment</i>	<i>Command</i>	<i>Output</i>
	number of iterations.		
7	To view the root for variable x.	[R/S]	2.00000
8	To view the root for variable y.	[R/S]	3.00000

## Memory Map

**R0 = x**  
**R1 = y**  
**R2 = hx, hy, J**  
**R3 = f(x, y)**  
**R4 = g(x, y)**  
**R5 = Fx(x, y)**  
**R6 = Fy(x, y)**  
**R7 = Gx(x, y)**  
**R8 = Gy(x, y)**  
**R9 = Iter**  
**RA = Toler x**  
**RB = Toler y**  
**RC = Delta x**  
**RD = Delta y**  
**RE = MaxIter**

## Listing

<i>Program Step</i>	<i>Comment</i>
◆ LBL A	Enter values for initial guesses for x and y.
STO 1	
X<>Y	
STO 0	
RTN	
◆ LBL a	Enter values for initial guesses for the tolerances of x and y.
STO A	
X<>Y	
STO E	
RTN	

<i>Program Step</i>	<i>Comment</i>
◆LBL B	Enter the maximum number of iterations and proceed with the iterations.
STO D	
0	
STO 9	I = 0
◆LBL 0	Start the main loop
RCL 1	
RCL 0	
PAUSE	Pause to display x
GSB 9	Calculate f(x,y)
STO 3	Store f(x,y)
RCL 1	
RCL 0	
GSB 8	Calculate g(x,y)
STO 4	Store g(x,y)
RCL 0	
ABS	
1	
STO+ 9	I = I + 1
+	
.001	
*	
STO 2	$h = 0.001 * (1 + \text{ABS}(x))$
RCL 1	
RCL 0	
RCL 2	
+	
GSB 9	Calculate f(x+h,y)
RCL 3	
-	
RCL 2	
/	
STO 5	Store Fx(x,y)
RCL 1	
RCL 0	
RCL 2	
+	

<i>Program Step</i>	<i>Comment</i>
GSB 8	Calculate $g(x+h,y)$
RCL 4	
-	
RCL 2	
/	
STO 7	Store $G_x(x,y)$
RCL 1	
PAUSE	Pause to display y
ABS	
1	
+	
.001	
*	
STO 2	$h = 0.001 * (1 + ABS(y))$
RCL 1	
+	
RCL 0	
GSB 9	Calculate $f(x,y+h)$
RCL 3	
-	
RCL 2	
/	
STO 6	Store $F_y(x,y)$
RCL 1	
RCL 2	
+	
RCL 0	
GSB 8	Calculate $g(x,y+h)$
RCL 4	
-	
RCL 2	
/	
STO 8	Store $G_y(x,y)$
RCL05	
*	
RCL 6	
RCL 7	

<i>Program Step</i>	<i>Comment</i>
*	
X<>Y	
-	
STO 2	$J = f_y * g_x - f_x * g_y$
1/X	Put 1/J in the stack
RCL 6	
RCL 4	
*	
RCL 3	
RCL 8	
*	
-	
*	Mutliply $(f_y * g - f * g_y)$ by 1/J
STO B	$DiffX = (f_y * g - f * g_y) / J$
STO- 0	$X = X - DiffX$
RCL 7	
RCL 3	
*	
RCL 4	
RCL 5	
*	
-	
RCL 2	
/	
STO C	$DiffY = (g_x * f - g * f_x) / J$
STO- 1	$Y = Y - DiffY$
RCL D	
RCL 9	
X>Y?	Is Iter > MaxIter?
GTO 1	Exit loop
RCL E	
RCL B	
ABS	
X>Y?	Is ABS(DiffX) > TolerX
GTO 0	Resume the next iteration
RCL A	
RCL C	

<i>Program Step</i>	<i>Comment</i>
ABS	
X>Y?	Is ABS(DiffY) > TolerY
GTO 0	Resume the next iteration
◆LBL 1	Display results
RCL 9	
R/S	Display the number of iterations
RCL 0	
R/S	Display root for variable x
RCL 1	
R/S	Display root for variable y
RTN	
◆LBL 9	Function f(x,y)
X^2	
X<>Y	
3	
Y^X	
+	
31	
-	
RTN	
◆LBL 8	Function g(x,y)
*	
6	
-	
RTN	